# - Returns a version string from Git
#
# These functions force a re-configure on each git commit so that you can
# trust the values of the variables in your build system.
#
#  get_git_head_revision(<refspecvar> <hashvar> [<additional arguments to git describe> ...])
#
# Returns the refspec and sha hash of the current head revision
#
#  git_describe(<var> [<additional arguments to git describe> ...])
#
# Returns the results of git describe on the source tree, and adjusting
# the output so that it tests false if an error occurs.
#
#  git_get_exact_tag(<var> [<additional arguments to git describe> ...])
#
# Returns the results of git describe --exact-match on the source tree,
# and adjusting the output so that it tests false if there was no exact
# matching tag.
#
# Requires CMake 2.6 or newer (uses the 'function' command)
#
# Original Author:
# 2009-2010 Ryan Pavlik <rpavlik@iastate.edu> <abiryan@ryand.net>
# http://academic.cleardefinition.com
# Iowa State University HCI Graduate Program/VRAC
#
# Copyright Iowa State University 2009-2010.
# Distributed under the Boost Software License, Version 1.0.
# (See accompanying file LICENSE_1_0.txt or copy at
# http://www.boost.org/LICENSE_1_0.txt)

if(__get_git_revision_description)
	return()
endif()
set(__get_git_revision_description YES)

# We must run the following at "include" time, not at function call time,
# to find the path to this module rather than the path to a calling list file
get_filename_component(_gitdescmoddir ${CMAKE_CURRENT_LIST_FILE} PATH)

function(get_git_head_revision _refspecvar _hashvar)
	set(GIT_PARENT_DIR "${CMAKE_SOURCE_DIR}")
	if (${ARGC} GREATER 2)
		list (GET ARGV 2 GIT_PARENT_DIR)
	endif()
	set(GIT_DIR "${GIT_PARENT_DIR}/.git")
	while(NOT EXISTS "${GIT_DIR}")	# .git dir not found, search parent directories
		set(GIT_PREVIOUS_PARENT "${GIT_PARENT_DIR}")
		get_filename_component(GIT_PARENT_DIR ${GIT_PARENT_DIR} PATH)
		if(GIT_PARENT_DIR STREQUAL GIT_PREVIOUS_PARENT)
			# We have reached the root directory, we are not in git
			set(${_refspecvar} "GITDIR-NOTFOUND" PARENT_SCOPE)
			set(${_hashvar} "GITDIR-NOTFOUND" PARENT_SCOPE)
			return()
		endif()
		set(GIT_DIR "${GIT_PARENT_DIR}/.git")
	endwhile()
	set(GIT_DATA "${CMAKE_CURRENT_BINARY_DIR}/CMakeFiles/git-data")
	if(NOT EXISTS "${GIT_DATA}")
		file(MAKE_DIRECTORY "${GIT_DATA}")
	endif()
	IF (NOT IS_DIRECTORY "${GIT_DIR}")	# special treatment for submodules / worktrees
		FILE(READ "${GIT_DIR}" NEW_GIT_DIR OFFSET 8)
		STRING(REGEX REPLACE "\n$" "" NEW_GIT_DIR "${NEW_GIT_DIR}")
		get_filename_component(GIT_DIR "${NEW_GIT_DIR}"
			REALPATH BASE_DIR "${GIT_PARENT_DIR}")
	ENDIF()
	if(NOT EXISTS "${GIT_DIR}/HEAD")
		MESSAGE(STATUS "GIT: Head does not exist in ${GIT_DIR}!")
		return()
	endif()
	set(HEAD_FILE "${GIT_DATA}/HEAD")
	configure_file("${GIT_DIR}/HEAD" "${HEAD_FILE}" COPYONLY)
	configure_file("${_gitdescmoddir}/GetGitRevisionDescription.cmake.in"
		"${GIT_DATA}/grabRef.cmake"
		@ONLY)
	include("${GIT_DATA}/grabRef.cmake")

	set(${_refspecvar} "${HEAD_REF}" PARENT_SCOPE)
	set(${_hashvar} "${HEAD_HASH}" PARENT_SCOPE)
endfunction()

function(git_describe out_nice out_shorthash)
	if(NOT GIT_FOUND)
		find_package(Git QUIET)
	endif()
	set(EXTRA_ARG "")
	if (${ARGC} GREATER 2)
		list (GET ARGV 2 EXTRA_ARG)
	endif()
	set(MYPATH ${CMAKE_SOURCE_DIR})
	if (${ARGC} GREATER 3)
		list (GET ARGV 3 MYPATH)
	endif()
	get_git_head_revision(refspec hash ${MYPATH})
	if(NOT GIT_FOUND)
		set(${out_nice} "GIT-NOTFOUND" PARENT_SCOPE)
		set(${out_shorthash} "GIT-NOTFOUND" PARENT_SCOPE)
		return()
	endif()
	if(NOT hash)
		set(${out_nice} "GIT-HEAD-HASH-NOTFOUND" PARENT_SCOPE)
		set(${out_shorthash} "GIT-HEAD-HASH-NOTFOUND" PARENT_SCOPE)
		return()
	endif()

	execute_process(COMMAND
		"${GIT_EXECUTABLE}" describe ${hash} ${EXTRA_ARG}
		WORKING_DIRECTORY "${MYPATH}"
		RESULT_VARIABLE res
		OUTPUT_VARIABLE	out
		ERROR_VARIABLE err
		OUTPUT_STRIP_TRAILING_WHITESPACE)
	if(NOT res EQUAL 0)
		MESSAGE(WARNING "Problem executing `git describe ${hash} ${EXTRA_ARG}`: ${err}")
		set(out "GIT-VERSION-NOTFOUND")
	else()
		IF ("${out}" MATCHES "[a-zA-Z0-9.]*-[0-9]*-[0-9a-z]*")
			STRING (FIND "${out}" "-" LASTHYPHENPOS REVERSE)
			MATH(EXPR SHORTHASHSTART "${LASTHYPHENPOS}+2")
			STRING(SUBSTRING "${out}" ${SHORTHASHSTART} -1 shorthash)
			STRING(SUBSTRING "${out}" 0 "${LASTHYPHENPOS}" substrout)
			STRING(FIND "${substrout}" "-" LASTHYPHENPOS REVERSE)
			STRING(SUBSTRING "${substrout}" 0 "${LASTHYPHENPOS}" versionstr)
			execute_process(COMMAND
				"${GIT_EXECUTABLE}" rev-parse --abbrev-ref HEAD
				WORKING_DIRECTORY "${MYPATH}"
				RESULT_VARIABLE res
				OUTPUT_VARIABLE	branch
				ERROR_VARIABLE err
				OUTPUT_STRIP_TRAILING_WHITESPACE)
			if(NOT res EQUAL 0)
				MESSAGE(WARNING "Problem getting current branch (through executing `git rev-parse --abrev-ref HEAD`): ${err}")
				set(branch "GIT-BRANCH-NOTFOUND")
			endif()
			STRING(CONCAT out "${versionstr}" "-" "${branch}" "-" "${shorthash}")
		ENDIF()
	endif()
	set(${out_nice} "${out}" PARENT_SCOPE)
	execute_process(COMMAND
		"${GIT_EXECUTABLE}" rev-parse --short ${hash}
		WORKING_DIRECTORY "${MYPATH}"
		RESULT_VARIABLE res2
		OUTPUT_VARIABLE	revparseout
		ERROR_VARIABLE err
		OUTPUT_STRIP_TRAILING_WHITESPACE)
	if(NOT res2 EQUAL 0)
		MESSAGE(WARNING "Problem executing `git rev-parse --short ${hash}`: ${err}")
		set(revparseout "GIT-SHORTHASH-NOTFOUND")
	endif()
	set(${out_shorthash} "${revparseout}" PARENT_SCOPE)
endfunction()

function(git_get_exact_tag _var)
	git_describe(out --exact-match ${ARGN})
	set(${_var} "${out}" PARENT_SCOPE)
endfunction()